######################################################
##
## EXERCISE 7A
##
######################################################

## Who is Caroline Kennedy’s father, and what is his birthdate and his death date?
#
SELECT ?father ?bdate ?ddate
WHERE {
  kennedys:CarolineKennedy schema:parent ?f .
  ?f schema:gender "male" .
  ?f schema:birthDate ?bdate .
  ?f schema:deathDate ?ddate .
}

## Who is Caroline Kennedy’s father, and what triples do we have in the graph that have this resource as subject?
#
SELECT *
WHERE {
  kennedys:CarolineKennedy schema:parent ?f .
  ?f schema:gender "male" .
  ?f ?p ?o.
}

## Who is whose mother
#
SELECT *
WHERE {
  ?child schema:parent ?mother .
  ?mother schema:gender "female" .
}

## Which persons graduated at the same university or college as one of their parents? 
#
SELECT *
WHERE {
  ?s schema:alumniOf ?org .
  ?s schema:parent ?parent .
  ?parent schema:alumniOf ?org .
} 

## Which persons have the same given name as one of their parents? 
#
SELECT *
WHERE {
  ?s schema:givenName ?name .
  ?s schema:parent ?parent .
  ?parent schema:givenName ?name .
}

## Which persons have the same given name as one of their grandparents? 
#
SELECT *
WHERE {
  ?s schema:givenName ?name .
  ?s schema:parent ?parent .
  ?parent schema:parent ?grandparent.
  ?grandparent schema:givenName ?name .
}

## Which persons have the same job title as one of their parents? 
#
SELECT *
WHERE {
  ?person schema:parent ?parent .
  ?person schema:jobTitle ?title .
  ?parent schema:jobTitle ?title .
}

## Which persons are married to a spouse that has the same job title as one of their parents?
#
SELECT *
WHERE {
  ?s schema:spouse ?spouse .
  ?spouse schema:jobTitle ?title .
  ?s schema:parent ?parent .
  ?parent schema:jobTitle ?title . 
}

##Find all dead Kennedys (assuming this to mean: with a specified death date)
#
SELECT ?person
WHERE {
  ?person schema:deathDate ?deathdate .
} 

######################################################
##
## EXERCISE 7B
##
######################################################

## Which distinct predicates are used in the Kennedys-graph?
#
SELECT DISTINCT ?p  
WHERE {
  [] ?p [] .
} 

##Which predicates are used with instances of the class schema:Person as object? 
#
SELECT DISTINCT ?p
WHERE {
  [] ?p ?o .
  ?o a schema:Person .
}

######################################################
##
## EXERCISE 7C
##
######################################################

##Create a list of all job titles that occur in the graph, sorted in ascending order
#
SELECT DISTINCT ?title 
WHERE {
  [] schema:jobTitle ?title.
} ORDER BY ASC(?title)

##Find the five persons with the oldest birthdates
#
SELECT DISTINCT ?person ?birthdate
WHERE {
  ?person schema:birthDate ?birthdate .
} ORDER BY ASC(?birthdate) LIMIT 5

##Find the two persons with the most recent death dates
#
SELECT ?person ?deathdate
WHERE {
  ?person schema:deathDate ?deathdate .
} ORDER BY DESC(?deathdate) LIMIT 2


## How many people graduated at Harvard
#
SELECT *
WHERE {
  ?s schema:alumniOf kennedys:Harvard.
}

## Find for each person which job titles that person has, but select only persons that have “navy” among their job titles
#
SELECT *
WHERE {
  ?person schema:jobTitle "navy".
  ?person schema:jobTitle ?title.
}

######################################################
##
## EXERCISE 7D
##
######################################################

## Find the labels of all superclasses of schema:CollegeOrUniversity excluding this class
#
SELECT ?c 
WHERE {
  schema:CollegeOrUniversity rdfs:subClassOf* / rdfs:label ?c .
} 

## Find all ancestors of Rory Kennedy
#
SELECT *
WHERE {
  kennedys:RoryKennedy schema:parent+ ?ancestor.
} 

## Find all offspring of Joseph Kennedy, ordered by birthdate (ascending)
#
SELECT *
WHERE {
  ?s schema:parent+ kennedys:JosephKennedy.
  ?s schema:birthDate ?date .
} ORDER BY ?date

## Find all subclasses of schema:Organization, including schema:Organization itself 
#
SELECT *
WHERE {
  ?class rdfs:subClassOf* schema:Organization .  
}

## Find all predicates that relate an instance of person to an instance of a subclass of schema:Organization.
#
SELECT DISTINCT ?p
WHERE {
  ?person a schema:Person .
  ?person ?p ?o .
  ?o a / rdfs:subClassOf* schema:Organization .
}

## Which persons are each other’s nephew and niece? 
#
SELECT DISTINCT  ?nephew ?niece
WHERE {
  ?nephew schema:parent / schema:parent ?grandparent  .
  ?niece  schema:parent / schema:parent ?grandparent .
  ?nephew schema:gender "male" .
  ?niece schema:gender "female" .
}

## The following query was used in the video
#
SELECT (COUNT (DISTINCT (?p)) AS ?c) 
WHERE {
  [] ?p [] .
} 

